
1  Abstract 
The TRS-80 MC-10 Micro Color Computer was released in 1983 as a competitor to the Timex Sinclair 
1000 and made available such features as audio and color video to the low-end home computer market.  
The architecture of the MC-10 appears unsophisticated when compared to its cousin released a few 
years earlier, the Tandy Color Computer.  Because of these design decisions, some features used by 
modern hardware upgrades are not as easily accessible as they are on slightly more advanced systems.  
On the MC-10, one way to deal with the missing video vertical sync information generally provided by 
the video display generator to the CPU is to synchronize its programmable timer to the vertical blanking 
region.  Doing so can provide access to improved video quality and enable access to CoCoVGA’s 
enhanced modes. 

2  Introduction 
Although the MC-10 is in the same Motorola-IC-based family as the Color Computer, it is considerably 
simpler.  Both the Color Computer and MC-10 share the same video controller chip, and this is evident 
when first connecting to a TV and powering it on; the highly recognizable green text display of the 6847 
video display generator (VDG) appears on the display.  However, examination of a physical system or 
schematic reveals that other ICs changed or were removed altogether.  For example, the 6883 
synchronous address multiplexer (SAM) is absent in the MC-10 and instead of the 6809E 
microprocessor, a more-integrated 6803 was employed.   

Some of these architectural decisions in the MC-10 result in limitations in capability compared to other 
home computers available at the time.  One feature, often desirable in graphical applications and 
games, is the display generator’s vertical sync interrupt, which indicates the raster has reached the end 
of the visible portion of the display (typically the bottom of the screen) and is preparing to start at the 
top of the display again after a period known as the vertical blanking region.  This interrupt, therefore, is 
periodic and matches the vertical refresh rate of the display.  On typical home computers in North 
America which use the NTSC standard, it is 60Hz, whereas European PAL displays are 50Hz.  
Programmers can use this information to perform display updates without “tearing”, which is a visual 
glitch in the display showing some of the updated frame on a portion of the upper part of the display 
and the previous video frame on the remainder.   

Because the MC-10 employs the same VDG as the CoCo, users who wish to install a CoCoVGA upgrade 
may do so by desoldering their 6847 and replacing it with a 40-pin socket to which a CoCoVGA board can 
connect.  This provides the base features needed to display standard MC-10 video on a VGA display.  
Some simple enhancements such as text screen color changes are accessible by additional button 
controls provided with CoCoVGA.   

Most CoCoVGA enhanced features are accessed via software which requires a specific orchestration of 
toggling the 5 VDG control bits (A/G, GM[2:0], and CSS) in a particular pattern during the vertical 



blanking region.  On a Color Computer, this would be followed by updating the SAM to point to a video 
page containing register or other information to CoCoVGA that it could consume as data instead of 
video.  During this video frame’s time, the previous double-buffered video is displayed by CoCoVGA.  
Shown below is a representation of how communication with CoCoVGA to access enhanced modes 
works on the Color Computer: 

 

Because of this requirement, it is not immediately obvious how one could reliably access this on an MC-
10, given that the vertical sync signal from the 6847 VDG is in no way connected to or available within a 
status register accessible by the CPU.   

3  Current State of the Art 
The MC-10’s Motorola 6803 microprocessor is highly integrated and brings with it such features as a 
programmable timer and built-in UART.  Thanks to efforts by Darren Atkinson, Simon Jonassen, and John 
Linville, a method by which to train the timer in the MC-10 to match what would be provided as a 
vertical sync interrupt has been demonstrated and provides a potential solution to this problem.   

Although the 6803 microcontroller brings with it a UART which relies on the timer, due to the common 
clock frequency needed for NTSC video used by the CPU, VDG, and video modulator, it is unused in the 
MC-10 in favor of a software-driven “bit-banged” serial interface which provides standard baud rates.  
This means that it is even more likely that the timer can be made available for this purpose without 
getting in the way of other hardware features that might otherwise be desirable to use. 



Note that for convenience, the code examples below use the following state values stored in the 
gocombolock variable.  During user synchronization, it is unused and set to 0.  Afterwards, however, 
it is used to signal the ISR (Interrupt Service Routine) that specific functionality is desired during that 
invocation as well as being used as a semaphore or handshake back to the main program regarding 
status: 

    ; 0 = disabled 
    ; 1 = combo lock requested during vsync ISR 
    ; 2 = combo lock issued, wait 1 display time 
    ; 3 = combo lock request handled by ISR, CoCoVGA regs updated 
 

From a high level, the technique demonstrated by the above authors is as follows: 

1. Enable the 6803 timer to run at the same vertical refresh frequency as the system and trigger 
interrupts each time it expires. 

2. Configure the timer to modify some user-visible attribute of the display for a limited portion of 
the above time period.  The authors above chose to use the CSS (color set select) which picks 
between green and orange text. 

3. Instruct the user to use keypresses to move the video region using keyboard controls so it is 
either the only region visible or so that it is not visible (i.e. in the blanking region).  
(gocombolock = 0) 

4. Keep the timer running and use its interrupt to trigger other software functionality that would 
ideally happen within the blanking region.  (gocombolock != 0) 

Below is a code listing which demonstrates an implementation of the steps above.  Both the initial 
configuration of the 6803 timer and ISR by inittocisr and the ISR itself, tocisr, are shown here: 

 
NTSCCPF EQU 14934 /* NTSC cycles-per-frame */ 
TCSR    EQU $0008 /* Timer Control and Status Register */ 
TIMER   EQU $0009 /* Timer Counter Register */ 
TOCR    EQU $000B /* Timer Output Compare Register */ 
 
TOCISRHOOK EQU $4206 /* Per disassembly listing http://www.roust-it.dk/coco/mc10/romlist.txt */ 
 
/* Init Timer ISR in RAM hook address */ 
 
inittocisr 

SEI   ; Mask interrupts 
  

CLR  TCSR  ; Clear Timer Control, disabling timer interrupts 
LDD  #NTSCCPF 
ADDD TIMER  ; Calculate Output Compare match NTSC-frame-size cycles in the 

;  future 
STD  TOCR  ; Save Output Compare value 

  
LDX  #tocisr 
STX  TOCISRHOOK+1  ; Save ISR address 
LDAA #$7E  ; JMP with 16-bit address 
STAA TOCISRHOOK 

 
LDAA #$08  ; Enable Output Compare interrupt 
STAA TCSR 

  
CLI   ; Enable interrupts 

  
RTS 

  



P1DDR EQU $0000 /* P1 Data Direction Register */ 
P2DDR EQU $0001 /* P2 Data Direction Register */  
P1DATA EQU $0002 /* P1 Data Register */ 
P2DATA EQU $0003 /* P2 Data Register */ 
 
WKEY  EQU $57 
ZKEY   EQU $5A 
ENTERKEY EQU $0D 
  
KVSPRT   EQU $BFFF /* Keyboard row data (read bits 5:0) + 
                   VDG mode data (write bits 6:2) + 
       1-bit sound data (write bit 7) */ 
 
/* Up/Down/Enter Pressed? function  */ 
/* Returns ASCII W, Z, or <CR> in A */  
updownenter 
 LDAA #$FB 
 STAA P1DATA 
 LDAA KVSPRT 
 BITA #$08 
 BEQ  zkey 
 
 LDAA #$BF 
 STAA P1DATA 
 LDAA KVSPRT 
 BITA #$08 
 BEQ  enterkey 
 
 LDAA #$7F 
 STAA P1DATA 
 LDAA KVSPRT 
 BITA #$04 
 BEQ  wkey 
 
 CLRA 
 BRA  updownenterdone 
 
zkey 
 LDAA #ZKEY 
 BRA  updownenterdone 
 
enterkey 
 LDAA #ENTERKEY 
 BRA  updownenterdone 
 
wkey 
 LDAA #WKEY 
 
updownenterdone 
 RTS 
 
 
 
/* Timer Output Compare ISR */ 
 
tocisr 
    ; There's no need to read TIMER – we are here due to an IRQ 
 LDD  TOCR  ; Get timer value at which interrupt occurred! 
 STD  lasttocr  ; Store for later 
  
/* Check to see if a combo lock request is set */ 
 LDAA gocombolock 
 CMPA #1 
 BEQ  tocisrcombo 
 CMPA #2 
 BEQ  tocccvupdated 
 BRA  tocisrkey 
 
/* One display time has elapsed, notify requester that CoCoVGA regs updated */ 
tocccvupdated 
 LDD  lasttocr 



 ADDD #NTSCCPF  ; Calculate future Output Compare to match NTSC-frame-size cycles  
 BITA TCSR  ; Clear Output Compare flag by doing any read (BITA is cheap load) 
 STD  TOCR  ; Save Output Compare value 
 
 LDAA #3 
 STAA gocombolock ; Set semaphore to notify that ISR has written combo lock and 
                              ; allowed CoCoVGA to read registers 
 JMP  tocisrdone 
  
  
/* COMBO LOCK PROGRAMMING DURING BLANKING REGION */ 
tocisrcombo 
 LDD  lasttocr 
 ADDD #NTSCCPF  ; Calculate Output Compare match NTSC-frame-size cycles in the 
future 
 BITA TCSR  ; Clear Output Compare flag by doing any read (BITA is cheap load) 
 STD  TOCR  ; Save Output Compare value 
 
; LDAB #$80  ; Wait into blanking region...  Doesn't work 
; LDAB #$60  ; Wait into blanking region...  Works! 
 LDAB #$40  ; Wait into blanking region...  Works! - pick "center" of working 

;   timing vsync range 
; LDAB #$20  ; Wait into blanking region...  Works! 
; CLRB   ; Wait into blanking region...  Doesn't work 
tocblankwait 
 LDAA #1   ; Get ready to multiply B by 1 to preserve B countdown 
 MUL   ; Waste time 
 DECB 
 BNE  tocblankwait ; Wait until B reaches 0 
 
 LDAA KVSPRT 
 ANDA #$83  ; Mask off bits to change 
 ORAA #$30  ; Set combo lock 1 
 STAA KVSPRT  ; Send to VDG socket 
  
 ANDA #$83  ; Mask off bits to change 
 ORAA #$44  ; Set combo lock 2 
 STAA KVSPRT  ; Send to VDG socket 
  
 ANDA #$83  ; Mask off bits to change 
 ORAA #$28  ; Set combo lock 3 
 STAA KVSPRT  ; Send to VDG socket 
 
 ANDA #$83  ; Mask off bits to change 
 ORAA #$7C  ; Set combo lock 4 
 STAA KVSPRT  ; Send to VDG socket 
 
 ANDA #$83  ; Mask off bits to change, no need to OR any bits in, want 0 
 STAA KVSPRT  ; Send to VDG socket 
 
 LDAA #2 
 STAA gocombolock ; Set semaphore to notify that ISR has written combo lock 
  
 BRA  tocisrdone 
  
  
/* WATCHING FOR USERS KEYPRESSES TO MOVE ORANGE STRIPE AROUND TO SYNCHRONIZE TIMER */ 
/* Look for user's keypresses */ 
tocisrkey 
 JSR  updownenter 
 CMPA #0 
 BEQ  tocisrskipdelay ; No key pressed, skip down to draw orange CSS stripe in  

;   same location 
  
 CMPA #ZKEY 
 BNE  checkwkey 
 
 LDD  #60  ; W (down-arrow) key pressed - add to move orange CSS stripe later! 
 ADDD lasttocr 
 STD  lasttocr 
 BRA  tocisrskipdelay 



  
checkwkey 
 CMPA #WKEY 
 BNE  tocisrskipdelay ; Some unrecognized key pressed, skip down to draw orange CSS  

;   stripe 
  
 LDD  #($FFFF-60) ; Z (up-arrow) key pressed - subtract to move orange CSS stripe  

;   earlier! 
 ADDD lasttocr 
 STD  lasttocr 
  
tocisrskipdelay 
 LDD  lasttocr 
 ADDD #NTSCCPF  ; Calculate future Output Compare to match NTSC-frame-size cycles  
 BITA TCSR  ; Clear Output Compare flag by doing any read (BITA is cheap load) 
 STD  TOCR  ; Save Output Compare value 
 
 LDAA #$40  ; Change to alternate (orange) CSS 
 STAA KVSPRT 
 
 LDAB #$E0  ; Draw band of orange CSS on screen by waiting... 
tocisrwait 
 LDAA #1   ; Get ready to multiply B by 1 to preserve B countdown 
 MUL   ; Waste time 
 DECB 
 BNE  tocisrwait ; Wait until B reaches 0 
 
 CLRA   ; Change back to green CSS 
 STAA KVSPRT 
  
tocisrdone  
 RTI 

 

In inittocisr, the two stores place a JMP opcode followed by a 16 bit address into 3 sequential 
bytes starting at $4206, which is where a stock MC-10 ROM’s vector pointer references. 

Both inittocisr and the ISR itself, tocisr, employ a constant CPU cycles-per frame value of 14934 
(NTSCCPF).  This is derived from 57 CPU cycles per video scanline (1/4 of the 228 NTSC “dot” or pixel 
clocks at a frequency of 3.5795MHz) by 262 scanlines total per frame.   

Note that to be completely correct, inittocisr should take the TIMER counter register value itself 
and add the NTSC CPU cycles-per-frame value minus the number of cycles needed to add the offset and 
store this in the Timer Output Compare Register (TOCR), however given how little time is spent dealing 
with this initial condition, this inaccuracy is invisible to the human eye.  In this case, any TOCR 
sufficiently in the future is acceptable. 

Obviously, the ISR consumes time in which the free-running timer continues to advance.  To avoid 
having to calculate the ISR entry latency and count the cycles necessary before the timer value register 
itself could be read to be subtracted from NTSCCPF, the last Timer Output Compare (TOCR) is used as 
the base to which to add this cycles-per-frame offset to select the timing of the next interrupt. 

As mentioned earlier, there are multiple modes of the ISR, whose execution path is dictated by 
gocombolock.  Note that each state updates TOCR with an appropriate value in the future.  During 
training (state 0), the increment value may vary, but in every other state, TOCR is always updated by 
adding NTSCCPF to it. 

 



State 0 – CoCoVGA-specific code disabled – user training 

In this state, the user is instructed to use the keyboard to move a region of the display with a particular 
color controlled by the CSS (Color Set Select) outside of the visible portion of the screen.  A majority of 
the ISR is spent checking for a user keypress (tocisrkey) and handling the CSS (Color Set Select) 
control of the 6847 (tocisrskipdelay/tocisrwait).  The keypresses recognized by 
updownenter are: 

 W (the up-arrow on CONTROL-shifted function) – subtracts from the previously stored TOCR in 
lasttocr to cause CSS video band to occur “earlier” on the display 

 Z (the down-arrow CONTROL-shifted function) – adds to the previously stored TOCR in 
lasttocr to cause CSS video band to occur “later” on the display 

 <ENTER> to accept that the video is correct per the on-screen instructions 

Note that in the above example code, handling of the <ENTER> key is handled by the main program 
loop instead of the ISR and is therefore not shown.  In other words, the main program loop moves 
gocombolock from state 0 to state 1 after performing any necessary intermediate steps as described 
later. 

In the segment of code following tocisrskipdelay/tocisrwait, the KVSPRT register which 
controls the 6847 pins including CSS, is manipulated and register B counts down from $E0 to 0, wasting 
significant time performing meaningless MULtiply operations.  This is non-destructive to B by loading a 
value of #1 into register A.  (A x B = 1 x B -> D places the value B back into the lower byte of D which 
contains B.)  While this timer is counting down, the video display changes from green to orange, and 
when the timer expires, the 6847 CSS is changed back to green. 

State 1 – enter CoCoVGA combo lock code sequence during blanking region 

Once the user has pressed the <ENTER> key to signify that the timer is properly synchronized to the 
video display, the main program may place appropriate values in video RAM to be used to program 
CoCoVGA registers or character RAM.  Since there is no way to remap video memory as in the Color 
Computer with its 6883 (SAM), it’s impossible on the MC-10 to compose this information on a separate 
video page, so the one-and-only visible video page must be used. 

When gocombolock is in state 1, the next invocation of the timer ISR will cause tocisr to execute 
the tocisrcombo path.  Since the user has trained the timer to trigger at the start of the blanking 
period, a sufficient “center” of the blanking region window with margin in which to apply the CoCoVGA 
combo lock code was found experimentally.  Then the code simply steps through the appropriate 
KVSPRT register programming sequence to input the combination lock which instructs CoCoVGA that 
the video frame it is about to consume is not video but should be destined for its registers and RAMs. 

The last step of this specific ISR code flow is to advance to state 2 of gocombolock. 

State 2 – wait one display refresh period 

Since it takes exactly one frame time for CoCoVGA to consume data after entering its combination lock 
code, this state simply executes the tocccvupdated path which updates TOCR by NTSCCPF and 
advances to state 3. 



State 3 – CoCoVGA programming complete 

At this point, if there is no need to maintain synchronization with display refresh, an SEI operation may 
be used to mask the timer interrupt.  For the above proof-of-concept, this is the expected functionality, 
however, the ISR could be updated to continue to track the blanking region for future use within the 
program. 

4  Areas for Improvement 
As mentioned earlier, another feature missing in the MC-10 is that of the SAM, which enables on-the-fly 
video RAM page remapping.  On systems with this capability, the CPU may draw an image on multiple 
copies of the screen in memory, each one slightly different than the others.  Then, during subsequent 
vertical blanking periods, it can flip the SAM’s video page pointer such that the VDG will display a new 
copy of the image to perform quick and low-CPU-cost animation without tearing.  To deal with this 
situation as it pertains to CoCoVGA register set programming, it may be possible to augment the code 
further to very quickly place the register values of interest to CoCoVGA during the blanking region when 
the combo lock code is being entered and then to revert these changes within the subsequent blanking 
region such that they would then not be visible on the display.  (Note that for the single frame time that 
CoCoVGA register and RAM upload data is in video RAM that CoCoVGA redisplays the previous video 
screen which it has buffered.) 

Additionally, if software wishes to take advantage of display updates without tearing, then it should 
perform operations which manipulate video memory during the blanking region.  This might mean that 
gocombolock state 3 could be a general-purpose mode which performs the appropriate display 
updates for the game or application software.  The same is true in the case that additional CoCoVGA 
programming is desired, perhaps to palette-cycle or update the character set, for example. 

The vector table in upper memory is masked by the MCX-128 when in MCX BASIC mode.  To re-enable 
the use of the stock MC-10 ROM’s vector pointer to $4206 requires: 

LDAA #$02   ; Deal with MCX-128 and MCX-32SD interrupt vector table 
STAA $BF01  ; Should have no effect on stock MC-10 where vector table is in ROM 

 

prior to calling the inittocisr routine.  Instead, in the future, specific advantage could be taken of 
the MCX-128’s ability to move the vector pointer to a higher RAM location to enable a larger contiguous 
video RAM region which is not interrupted by the JMP and address at $4206.  In other words, the 
original MC-10’s vector table in ROM points into the region between $4000 and $43FF used by CG1, 
RG1, and all other non-semigraphics modes.  This means that on a stock MC-10, the programmer has the 
choice of: 

1. using timer interrupts and putting a valid JMP and address at $4206 and making this also visible 
within the graphics screen (as part of the background, perhaps) 

2. using timer interrupts and limiting software to the 512-byte text and semigraphics video mode 
which would then not collide with each other 

3. not using timer interrupts so that the entire video region can be used for graphics 



Providing an alternative to the above 3 options for users of the MCX-128 and MCX-32SD could be very 
advantageous to deal with the MC-10’s limitations in this respect.  Doing so would make use of the 
graphical video modes possible while still having interrupt-driven timer code present.   

5  Conclusions and Future Work 
In some situations, the training of the 6803’s timer may feel to the user to be a bit contrived.  If this is 
intended for use within a game, one possibility could be to “gamify” this training.  For example, early in 
the game, the region to move off the screen could gradually but continually drift downwards, and the 
user could bump it off the top of the screen using their player cursor by bouncing a projectile from it.  
Another option would be for the region to drift down the screen (with an opening) and then the user 
must navigate through the opening.  Detection of when and how the user reacts could be used to 
determine when the region of interest is in a particular position of the display. 

As we have seen, a small amount of user intervention and taking advantage of other hardware features 
can help compensate for design limitations and enable more complete use of modern hardware 
upgrades for the MC-10.  These, in turn, enable more complex and feature-rich software to continue to 
push these vintage home computer systems’ limits. 

